***************************************************************************************************************
* Name: TableA7
* Description: Content of Table is generated with final dataset created by running this file
* Reference: "Competition and Civilian Victimization"
***************************************************************************************************************

use "violence_t_year_export.dta",clear
*drop pred_victim_farc pred_victim_paras nbi_ts1-coca2
keep if victim_farc~=.&victim_paras~=.&nbi_t~=.&royalties_t~=.&coca_t~=.&share_left~=.&lpobl_tot_t~=.&army_dist~=.&dmr~=.&gini_i~=.&evlp~=.&gcaribe~=.&gpacifica~=.&gorinoquia~=.&gamazonia~=.&farc_dist~=.&paras_dist~=.&year~=.

*****Log likelihood definition*****

global vF "victim_farc"
global vP "victim_paras" 

*Victimization un-restricted
capture program drop lfsec_ur
program lfsec_ur
  version 13.1
  args lnf xbF xbP
  tempvar pF pP
  quietly {
     gen double `pF' = 1/(1+exp(-`xbF'))
     gen double `pP' = 1/(1+exp(-`xbP'))
     replace `lnf' = $vP* ln(`pP')+(1-$vP)*ln(1-`pP') + $vF*ln(`pF')+(1-$vF)*ln(1-`pF') 
   }
end


*Two stage estimation
 capture program drop game2stage
 program game2stage, eclass
	syntax [if]
	marksample touse
	tempname b b2 dF2 dP2
	tempvar predF predP
	local controls "nbi_t royalties_t coca_t share_left lpobl_tot_t year_d1-year_d7 army_dist gini_i dmr evlp gcaribe gpacifica gorinoquia gamazonia"
	
	quietly {
		npregress kernel victim_farc `controls' farc_dist if victim_paras~=., estimator(constant) noderivatives
		predict `predF'
		npregress kernel victim_paras `controls' paras_dist if victim_paras~=.,estimator(constant) noderivatives
		predict `predP'
	}
	
	gen pF=`predF'
	gen pP=`predP'
		
	corr  pF victim_farc pP victim_paras
	ml model lf lfsec_ur (eq1: victim_farc = nbi_t royalties_t coca_t share_left lpobl_tot_t army_dist  ///
						 dmr gini_i  evlp gcaribe gpacifica gorinoquia gamazonia farc_dist year_d1-year_d7 pP) (eq2: nbi_t royalties_t coca_t ///
						share_left lpobl_tot_t army_dist dmr gini_i  ///
						 evlp  gcaribe gpacifica gorinoquia gamazonia paras_dist year_d1-year_d7 pF)
	capture noisily ml maximize, diff 
	display "maximize" _rc
	if _rc~=0 {
		matrix `b2'=J(1,46,0)
		matrix `dF2'=J(1,13,0)
		matrix `dP2'=J(1,13,0)	
	}
	else{
		matrix `b2'=e(b)

		quietly{
			margins , at(pP=(0.2(0.05)0.8) year_d1=0 year_d2=0 year_d3=0 year_d4=0 year_d5=0 year_d6=0 year_d7=0 gcaribe=0 gpacifica=0 gorinoquia=0 gamazonia=0 ) atmeans expression(exp(predict(equation(eq1)))/(1+exp(predict(equation(eq1))))) nose
			matrix `dF2'=r(b)

			margins , at(pF=(0.2(0.05)0.8) year_d1=0 year_d2=0 year_d3=0 year_d4=0 year_d5=0 year_d6=0 year_d7=0 gcaribe=0 gpacifica=0 gorinoquia=0 gamazonia=0 ) atmeans expression(exp(predict(equation(eq2)))/(1+exp(predict(equation(eq2))))) nose
			matrix `dP2'=r(b)
			}
	}		
	matrix `b'=(`b2',`dF2',`dP2')		
		
	drop pF pP
	ereturn post `b' ,esample(`touse')		
	
end

game2stage
matrix b=e(b)
matrix res_boot=b
set seed 1313

set matsize 800

forvalues i = 1/500{
	preserve
*	bsample ,cluster(muni_code)
	bsample
	game2stage
	matrix b=e(b)
	matrix res_boot=(res_boot\b)
	display `i'
	restore
}

svmat double res_boot
keep res_boot1-res_boot72
save "boot_res_region_v_year.dta",replace
use "boot_res_region_v_year.dta",clear


forvalue i=1/72 {
	replace res_boot`i'=. if res_boot`i'==0
}	

tabstat res_boot1-res_boot72 if _n~=1,s(sd) save
matrix sd_mat=r(StatTotal)
keep if _n==1
gen dum=1
reshape long res_boot, i(dum) j(estimates)
drop dum estimates
rename res_boot estimates
matrix sd_mat=sd_mat'
svmat sd_mat

*Generating pvals and confidence intervals
gen pval=min(2*normal(estimates/sd_mat),2*(1-normal(estimates/sd_mat)))

drop if _n>=47

list sd_mat
list pval
